
Abstract

Substitution boxes (S-boxes) are generally the only non-linear part of a block cipher. Using
key-dependent S-boxes rather than static ones might increase the security of a block cipher
but it would take too much time to build a good key-dependent S-box from scratch just before
ciphering or deciphering. What we can do is to transform an existing S-box, assuming

1) the relevant cryptographic properties of the S-box are preserved;
2) the execution of the transformation is sufficiently fast.

1 Introduction

First, a few words to justify the title: the transformation is « easy » because its implementation
does not require particular math resources (no binary matrix library, for instance) and it is « fast »
because transforming an existing S-box takes only a few microseconds.

Notation used
Z2n The set of the integers 0..2n − 1, n > 0.
Zn
2 The set of the column vectors v = t(v0, v1, ..., vn−1) having n components in Z2, n > 0.
V(k) Vector k of Zn

2 equivalent to the integer k of Z2n [1].
I(v) Integer v of Z2n equivalent to the vector v of Zn

2 [2].
⊕ The xor operator. Bitwise exclusive or over Z2n as well as over Zn

2 (over Zn
2 , the ⊕ and +

operators are equivalent).
Mn

2 The set of the n×n matrices having their coefficients in Z2, n > 0.
∗ The multiplication operator of a matrix of Mn

2 by a vector of Zn
2 .

If u = M ∗ v then, ∀i ∈ 0..n− 1, ui =
⊕

0≤k<n

(Mi,k×vk).

A (n, n) S-box is a mapping S : Zn
2 → Zn

2 where n is a fixed positive integer, n ≥ 2. Such a
S-box can be represented as an array of 2n integers of Z2n [9].

Two (n, n) S-boxes R and S are said affine equivalent if R = B ◦ S ◦ A where A and B are
affine permutations. With affine permutations, many cryptographic properties of the transformed
S-boxes have the same values than the ones of the initial S-boxes. Among others, the balanced-
ness, the algebraic degree, the minimum degree, the non-linearity, the differential uniformity, the
algebraic immunity, and the global avalanche characteristics [11] are identical [3, 4, 5]. The graph
algebraic immunity indicators (degree and number of independent equations) [7] are also pre-
served [3].

From a computational point of view, in order to perform the affine transformation of a S-box, only
two permutations (look-up tables) P and Q are needed. As a matter of fact, the S-box being coded
as an array of 2n integers of Z2n , the proposed method simply consists in permuting the 2n indices
of this array according to P and, then, in permuting the 2n values according to Q.

In what follows, most algorithms (more exactly, Free Pascal functions) are specialized for (8, 8)
S-boxes but it is not difficult to translate them in an other language and/or to adapt them so that
they work with other dimensions.

For the Pascal code, the global type used is
type TArrayOfByte = array [Byte] of Byte; // array of 256 bytes

It is assumed that the NextByte() function, using a part of the secret key of a block cipher as a
seed, behaves like a random generator having a uniform distribution on Z28 .

1There is a bijective mapping between the integers of Z2n and the vectors of Zn
2 defined by the correspondence

(b0 + 2b1 + ...+ 2n−1bn−1) ∈ Z2n ←→ t(b0, b1, ..., bn−1) ∈ Zn
2 .

2See the previous note.
3The graph algebraic immunity is invariant under CCZ-equivalence [4] and the affine equivalence of S-boxes implies

their CCZ-equivalence [8, Slide 53].

2



2 Building of a permutation

Let M be an invertible matrix of M8
2 and let a be an integer of Z28 . It is easy to build a permutation

P of the Z28 integers such that, ∀k ∈ Z28 , P [k] = I(M ∗ k)⊕ a.

Let us first consider the case a = 0. Ci being the column vector #i of M , if we build P such that
P [2i] ← Ci (for i in 0..7),
P [i⊕ j] ← P [i]⊕ P [j] (for all the other cells of P ),

then, ∀k ∈ Z28 , P [k] = I(M ∗ k). (The proof is not difficult.)

For the case a 6= 0, it is sufficient to xor with a all the values of the previously obtained permuta-
tion P (but this is not the way it is coded in the following Pascal function).

// ——————————————————————————————————————
//→ a, addend. If a = 0 then Result is equivalent to a linear transformation.
//← Result, key-dependent permutation of the Z28 integers (based on an invertible 8×8 binary
// matrix).
// ——————————————————————————————————————
function GetKeyedSwap(a: Byte): TArrayOfByte;

var
P : TArrayOfByte absolute Result; // P is an alias for Result
FreeColumn : array [Byte] of Boolean;
i, j : Integer;
c, t : Byte;

begin
FillChar(FreeColumn,SizeOf(FreeColumn),TRUE);
FreeColumn[a] := FALSE;
P[0] := a;

// columns #0..#6
j := 1;
repeat

// get a new independent column
repeat

c := NextByte;
until FreeColumn[c xor a]; // proba: (256 − j)/256

// update P and mark all linear combinations of columns already used
for i := 0 to j − 1 do
begin

t := P[i] xor c;
P[i xor j] := t;
FreeColumn[t] := FALSE;

end;
j := j shl 1;

until j = 128;

// column #7 (no more need to update FreeColumn[])
// get a new independent column
repeat

c := NextByte;
until FreeColumn[c xor a]; // proba: 1/2

// update P (type-casting P on the fly to a DWord or a QWord array might speed things up)
for i := 0 to 127 do

P[i xor 128] := P[i] xor c;
end;
// ——————————————————————————————————————

Remarks

• By construction, if a = 0 then
{
P : Z28 → Z28

i 7→ P [i]

}
is an automorphism of the group (Z28 ,⊕).

• Assuming the NextByte() function behaves like a uniform random generator, on average, ob-
taining the eight independent columns requires

∑
0≤i<8

28

28 − 2i
≈ 9.603 calls to this function. So, on

average, in order to make P affine, there are only about 9.603 executions of the "xor a" instruction
(instead of 256 if a simple loop were used).

• There are N = 28×
∏

0≤i<8
(28 − 2i) = 1 369 104 324 918 194 995 200 [4] different possible

permutations. Compared to powers of 2, we have 270 < N < 271.

3 Elimination of fixed points

Suppose we intend to xor all the values of a S-box S with the value a. If (S[i]⊕ a = i) then i will
be a fixed point.
Now, since (S[i] ⊕ a = i) ⇐⇒ (S[i] ⊕ i = a), if we fill up a boolean array T with the value
FALSE and then if we set T [S[i] ⊕ i] to TRUE for all i, the cells T [j] remaining FALSE, if any,
indicate that xoring the S-box S with j will produce a direct fixed point free S-box. And we get
all the possible j’s in a raw.

// ——————————————————————————————————————
//↔ S, (8, 8) S-box.
//← Result, TRUE if, and only if, the modified S is (direct and opposite) fixed point free.
// ——————————————————————————————————————
function MakeFixedPointFree(var S: TArrayOfByte): Boolean;

var
FixedPoint : array [Byte] of Boolean;
i, j : Byte;

4(Number of different values a) × (Number of different invertible matrices of M8
2)

begin
FillChar(FixedPoint,SizeOf(FixedPoint),FALSE);
for i := 0 to 255 do
begin

j := S[i] xor i;
FixedPoint[j] := TRUE; // direct fixed point
FixedPoint[not j] := TRUE; // opposite fixed point: S[i] xor (not i) = not (S[i] xor i)

end;

i := NextByte; // key-dependent initial index
j := i;
while FixedPoint[j] do
begin

j := (j + 1) and $FF; // since j is a byte, a simple "Inc(j);" would be OK
if j = i then

Exit(FALSE); // no solution
end;

// modify S (type-casting S on the fly to a DWord or a QWord array might speed things up)
for i := 0 to 255 do

S[i] := S[i] xor j; // useless whenever j = 0

Result := TRUE;
end;
// ——————————————————————————————————————

Remarks
• The final FixedPoint array is palindromic since, with any byte j, not j = 255− j. Therefore
its size could be divided by 2 but all what I have tried led to a slower code.

• Though it is theoretically possible that MakeFixedPointFree() returns FALSE, it would seem it
rarely occurs with S-boxes having good cryptographic properties [5]. (As a matter of fact, during
the numerous tests I made, it never occurred.)

5With (8, 8) S-boxes: Minimum degree = 7, non-linearity ≥ 104 and differential uniformity ≤ 8.

4 Affine transformation of a S-Box

The affine transformation R = B ◦ S ◦ A, where R and S are S-boxes and A and B are affine
permutations, can be computed with Mn

2 matrices and additions of Zn
2 vectors (see [10, Slide 6]).

Using the permutations produced by GetKeyedSwap(), we can get the S-box R with

R[i]← PB[S[PA[i]]], ∀i ∈ 0..2n − 1.

// ——————————————————————————————————————
//→ S, (8, 8) S-box.
//← Result, key dependent (8, 8) S-box. Result is affine equivalent to S and it is direct (as well
// as opposite) fixed point free.
// ——————————————————————————————————————
function GetKeyedSBox(const S: TArrayOfByte): TArrayOfByte;

var
R : TArrayOfByte absolute Result; // R is an alias for Result
T : array [0..1] of TArrayOfByte;
P : TArrayOfByte absolute T[0];
Q : TArrayOfByte absolute T[1];
i, j : Integer;

begin
P := GetKeyedSwap(NextByte);
j := 0;
repeat

j := j xor 1;
T[j] := GetKeyedSwap(NextByte);
for i := 0 to 255 do

R[i] := Q[S[P[i]]];
until MakeFixedPointFree(R);

end;
// ——————————————————————————————————————

Remarks
• Since (R = B ◦ S ◦ A) ⇐⇒ (R ◦ A−1 = B ◦ S), we could also build the S-box R with
R[P−1

A [i]]← PB[S[i]], ∀i ∈ 0..2n − 1.

• Due to the symmetry properties of S (if any), even if P and Q are different from the Identity
permutation, R might be equal to S (see [2, §4.1, Self-Equivalent S-boxes]).

5 Examples of use

• Example 1

Table 1: Original AES S-box [6]
63 7c 77 7b f2 6b 6f c5 30 01 67 2b fe d7 ab 76
ca 82 c9 7d fa 59 47 f0 ad d4 a2 af 9c a4 72 c0
b7 fd 93 26 36 3f f7 cc 34 a5 e5 f1 71 d8 31 15
04 c7 23 c3 18 96 05 9a 07 12 80 e2 eb 27 b2 75
09 83 2c 1a 1b 6e 5a a0 52 3b d6 b3 29 e3 2f 84
53 d1 00 ed 20 fc b1 5b 6a cb be 39 4a 4c 58 cf
d0 ef aa fb 43 4d 33 85 45 f9 02 7f 50 3c 9f a8
51 a3 40 8f 92 9d 38 f5 bc b6 da 21 10 ff f3 d2
cd 0c 13 ec 5f 97 44 17 c4 a7 7e 3d 64 5d 19 73
60 81 4f dc 22 2a 90 88 46 ee b8 14 de 5e 0b db
e0 32 3a 0a 49 06 24 5c c2 d3 ac 62 91 95 e4 79
e7 c8 37 6d 8d d5 4e a9 6c 56 f4 ea 65 7a ae 08
ba 78 25 2e 1c a6 b4 c6 e8 dd 74 1f 4b bd 8b 8a
70 3e b5 66 48 03 f6 0e 61 35 57 b9 86 c1 1d 9e
e1 f8 98 11 69 d9 8e 94 9b 1e 87 e9 ce 55 28 df
8c a1 89 0d bf e6 42 68 41 99 2d 0f b0 54 bb 16

Some properties of the AES S-box:

• Balanced = TRUE
• Minimum degree = 7
• Non-linearity = 112
• Differential uniformity = 4
• Global avalanche characteristics:

– Absolute indicator = 32
– Sum-of-Squares indicator = 133120

• Graph algebraic immunity:
– Degree = 2
– Independent equations = 39

• Fixed points:
– Direct = 0
– Opposite = 0

Code used in order to make things easily reproducible:

var Seed : Byte;

function NextByte: Byte; inline;
begin

Seed := Seed*5 + 131;
Result := Seed;

end;

Seed := 1;
S := GetKeyedSBox(AES); // AES: array of 256 bytes

Table 2: S-box S returned by GetKeyedSBox(AES);
c5 49 d0 cc 88 07 a0 98 f2 59 14 61 ce f8 77 6f
2d b0 bc e1 95 eb 30 50 8d 71 3d 08 82 26 18 6a
f6 37 e5 7c 72 4d fe 93 40 7a ca 68 56 66 8a aa
f1 9b a6 97 cf 79 1d 21 bf 20 69 ea d1 a5 ae 24
df 1f a2 70 02 7b 3e fd 84 04 2f 46 e2 f4 ef 5e
22 90 76 87 f7 0f 62 48 cd 28 a4 1b 9e 67 f5 fc
bd 03 94 ec c0 2c e3 34 47 ac e7 c6 75 25 64 53
2a 5b 83 86 1c 89 4a 9f 60 c1 7f c4 81 57 6c 36
33 10 3c be 78 8e 4e 0c 6d 8b 5c 58 0a ad de 00
c8 29 3a b7 c2 6e 73 db c3 0b b2 52 f9 dc dd a7
cb e6 38 01 5a 15 d4 a3 d8 16 2e 63 23 17 d9 4b
d5 c7 1e a9 ee e4 19 31 b3 35 3b b5 42 91 92 32
5d e8 da 8c b6 85 11 99 0e 7d b9 09 ab f3 74 4f
2b af d6 b4 bb d3 80 13 a8 e0 b8 ba 55 c9 b1 27
8f 65 ff 9a d2 9c 51 96 9d 45 12 41 43 f0 d7 fb
44 54 6b a1 4c e9 ed 3f fa 0d 39 7e 05 1a 06 5f

All the previously listed properties have the same values for the modified S-Box than for the AES
S-box.
The average running time of the GetKeyedSBox() function, obtained with Free Pascal 3.0.4 on
a Intel i7-2600 processor (3.4 GHz), is equal to 3.3 µs.

• Example 2

In [1], the authors propose a method based on two small permutations to transform a S-box. Their
method being a linear transformation, the integer permutation trick proposed here can be used to
do the job.
In order to get the same results than the ones of their examples 3.5 & 3.7, we only need two
functions: GetSwap() and GetSBox().
The global type used is

type TDynArrayOfByte = array of Byte; // dynamic array of bytes

// ——————————————————————————————————————
//→ p, permutation of 0..n− 1, n ≤ 8. p is regarded as the representation of a n×n permutation
// matrix M such that Mp[i],i = 1.
//← Result, permutation of 0..2n − 1 equivalent to M , i.e., to p.
// ——————————————————————————————————————
function GetSwap(const p: array of Byte): TDynArrayOfByte;

var
n : Integer;
i, j, k, c : Byte;

begin
n := Length(p);
SetLength(Result,Integer(1) shl n);
Result[0] := 0;
i := 1;

for j := 0 to n − 1 do
begin

c := Byte(1) shl p[j]; // c is the column vector #j of a n×n permutation matrix
for k := 0 to i − 1 do

Result[k xor i] := Result[k] xor c;
i := i shl 1;

end;
end;
// ——————————————————————————————————————

Remarks
• The parameter of GetSwap() is declared as array of Byte (i.e., "open" array). This way the
function may be called with any array type: ordinary, open, or dynamic.

• Let R be the permutation returned by GetSwap() and let M be the n×n permutation matrix
equivalent to p (i.e., such that Mp[i],i = 1). Then, ∀k ∈ Z2n , R[k] = I(M ∗ k).
For instance, with n = 8, p = (1, 2, 0, 3, 5, 7, 6, 4) and k = 108, we get

M =



0 0 1 0 0 0 0 0
1 0 0 0 0 0 0 0
0 1 0 0 0 0 0 0
0 0 0 1 0 0 0 0
0 0 0 0 0 0 0 1
0 0 0 0 1 0 0 0
0 0 0 0 0 0 1 0
0 0 0 0 0 1 0 0


and

M ∗ k = M ∗ t(0, 0, 1, 1, 0, 1, 1, 0) = t(1, 0, 0, 1, 0, 0, 1, 1) = V(1 + 8 + 64 + 128) = V(201).
Of course, R[108] = 201.

// ——————————————————————————————————————
//→ S, (n, n) S-box (array of 2n bytes), n ≤ 8.
//→ p1, permutation of 0..n− 1 (array of n bytes).
//→ p2, permutation of 0..n− 1 (array of n bytes).
//← Result, S-box S modified according to the p1 and p2 permutations.
// ——————————————————————————————————————
function GetSBox(const S, p1, p2: array of Byte): TDynArrayOfByte;

var
P, Q : TDynArrayOfByte;
i : Byte;

begin
P := GetSwap(p1);
Q := GetSwap(p2);
SetLength(Result,Length(S));
for i := 0 to High(S) do

Result[i] := Q[S[P[i]]];
end;
// ——————————————————————————————————————

Remark
Except for the numbers of fixed points, the properties (listed in Example #1) of the returned
S-boxes have the same values than the ones of S.

(For the two following resulting S-boxes, I made use of a decimal representation to mimic what is
given in the quoted paper.)

As in the example 3.5, page 10 of [1], the (4, 4) S-box returned by

GetSBox([9,13,10,15,11,14,7,3,12,8,6,2,4,1,0,5], [1,2,0,3], [3,2,0,1]);

is equal to

10 6 14 13 11 15 7 12 3 5 1 0 2 4 8 9

As in the example 3.7, page 14 of [1], the (8, 8) S-box returned by

GetSBox(AES, [1,2,0,3,5,7,6,4], [1,0,2,3,7,5,4,6]);

is equal to

51 183 241 63 188 187 59 86 160 55 253 107 2 43 215 181
231 195 165 247 254 37 175 92 164 118 178 162 102 242 216 134
94 131 159 20 12 124 199 135 84 189 52 138 103 174 158 179
112 169 26 36 161 9 5 156 81 108 194 116 211 49 198 186
10 44 139 153 67 137 61 96 145 213 42 47 171 227 115 68
208 105 19 163 127 251 30 70 22 1 144 207 250 191 172 104
233 38 140 228 184 45 101 85 120 180 27 75 222 143 238 73
114 200 58 77 248 130 218 196 203 71 93 40 141 122 150 223
89 90 249 23 65 190 154 240 110 97 204 177 212 111 100 80
4 35 136 6 87 83 197 201 7 64 123 225 129 113 39 182

48 31 33 192 66 220 41 72 21 232 221 11 125 132 157 219
119 167 78 29 88 62 214 106 60 244 54 109 149 121 185 8
147 0 32 226 210 126 252 155 57 237 25 152 91 170 28 95
146 16 193 168 99 79 206 246 236 217 128 243 229 34 255 209
176 230 24 245 173 53 3 13 50 151 69 142 166 234 82 205
76 74 239 17 98 14 117 56 18 46 224 235 202 15 148 133

6 Conclusion

The proposed method is very easy to implement. From a practical point of view, it is sufficiently
light and fast so that it can be embedded in a block cipher.

It should be noted that affine transformations modify some cryptographic properties of a S-box.
Among others:

– the number of monomials of the univariate polynomial representation;
– the transparency order;
– the branch number;
– the DPA [6] signal-to-noise ratio.

These properties may become better... or the contrary.

***

Copyright © 2022, Marcel Martin
First publication October 18, 2020

6Differential Power Analysis.

11



References

[1] A Novel Method to Generate Key-Dependent S-Boxes with Identical Algebraic Proper-
ties
Ahmad Y. Al-Dweik, Iqtadar Hussain, Moutaz S. Saleh & M. T. Mustafa (2019)

[2] A Toolbox for Cryptanalysis: Linear and Affine Equivalence Algorithms
Alex Biryukov, Christophe De Cannière, Ann Braeken & Bart Preneel (Eurocrypt, 2003)

[3] On the behaviours of affine equivalent Sboxes regarding differential and linear attacks
Anne Canteaut & Joëlle Roué (Eurocrypt, 2015)

[4] On the Algebraic Immunities and Higher Order Nonlinearities of Vectorial Boolean
Functions
Published in "Enhancing Cryptographic Primitives with Techniques from Error Correcting
Codes", pp. 104-116.
Claude Carlet (2009)

[5] Vectorial Boolean Functions for Cryptography
Claude Carlet (2010)

[6] Advanced Encryption Standard - FIPS 197
FIPS (2001)

[7] On some methods for constructing almost optimal S-Boxes and their resilience against
side-channel attacks
Reynier Antonio de la Cruz Jiménez (2018)

[8] Analysing S-boxes
Gregor Leander (2012)

[9] Practical S-box Design
Serge Mister & Carlisle Adams (SAC, 1996)

[10] A Search Strategy to Optimize the Affine Variant Properties of S-boxes
Stjepan Picek, Bohan Yang & Nele Mentens (2016)

[11] GAC - the Criterion for Global Avalanche Characteristics of Cryptographic Functions
Xian-Mo Zhang & Yuliang Zheng (1997)

12


	Introduction
	Building of a permutation
	Elimination of fixed points
	Affine transformation of a S-Box
	Examples of use
	Conclusion

